Correction du TP n°7

Exercice 1
Attention, on demande des ordres au sens de votre cours de mathématiques. Il faut donc faire attention aux ordres que vous demandez à Maple et penser à la règle "Un $O(x^n)$ est un $o(x^{n-1})$".
Remarque : $x^n \varepsilon(x)$ (avec $\varepsilon(x) \to 0$) est un $o(x^n)$.

```
> restart;
```

1)

```
> taylor((ln(1+x))^2,x,7);
```

$$x^2 - x^3 + \frac{11}{12}x^4 - \frac{5}{6}x^5 + \frac{137}{180}x^6 + O(x^7) \quad (1)$$

Donc $(\ln(1+x))^2 = x^2 - x^3 + \frac{11\,x^4}{12} - \frac{5\,x^5}{6} + \frac{137\,x^6}{180} + o(x^6)$.

2)

```
> taylor(sqrt(x*(sin(x)+sinh(x)-2*x)),x,10);
```

$$\frac{\sqrt{15}}{30}x^3 + O(x^7) \quad (2)$$

Demander un calcul à l'ordre Maple 10 ne suffit pas pour obtenir un $o(x^9)$, donc on pousse plus loin :

```
> taylor(sqrt(x*(sin(x)+sinh(x)-2*x)),x,11);
```

$$\frac{1}{30}\sqrt{15}\,x^3 + \frac{1}{181440}\sqrt{15}\,x^7 + O(x^{11}) \quad (3)$$

Cela nous donne le résultat, puisque qu'un $O(x^{11})$ est un $o(x^9)$.

3)

```
> taylor(sin(x)/(1+x),x=Pi/2,4);
```

$$\frac{1}{1+\frac{1}{2}\pi} - \frac{2}{(2+\pi)\left(1+\frac{1}{2}\pi\right)}\left(x-\frac{1}{2}\pi\right) + \frac{-\frac{1}{2}+\frac{4}{(2+\pi)^2}}{1+\frac{1}{2}\pi}\left(x-\frac{1}{2}\pi\right)^2 + \frac{-4+4\pi+\pi^2}{(2+\pi)^3\left(1+\frac{1}{2}\pi\right)}\left(x-\frac{1}{2}\pi\right)^3 + O\left(\left(x-\frac{1}{2}\pi\right)^4\right) \quad (4)$$

4)

```
> taylor(1+2*x+3*x^5+x^6,x,6);
```

$$1 + 2x + 3x^5 + O(x^6) \quad (5)$$

Cela nous dit que $x^6 = o(x^5)$.

```
> taylor(1+2*x+3*x^5+x^6,x,7);
```

$$1 + 2x + 3x^5 + x^6 \quad (6)$$

On constate que le O a disparu.C'est normal : un polynôme a un développement limité exact (sans reste en o ni O), pourvu qu'on le calcule à un ordre suffisamment grand. Si on pousse plus loin, on obtient le même résultat :

```
> taylor(1+2*x+3*x^5+x^6,x,10);
```

$$1 + 2x + 3x^5 + x^6 \quad (7)$$

5) Ici, pas la peine de fixer une valeur pour *a*.

```
> taylor((1+x)^a,x,6);
```

$$1 + a\,x + \frac{1}{2}a(a-1)x^2 + \frac{1}{6}a(a-1)(a-2)x^3 + \frac{1}{24}a(a-1)(a-2)(a-3)x^4 + \frac{1}{120}a(a-1)(a-2)(a-3)(a-4)x^5 + O(x^6) \quad (8)$$

Exercice 2
Rappel mathématique : On dit que f est équivalente à g en 0 si $g - f = o(f)$.

Calculons un développement limité de g en 0.

```
> restart;
  g:=x->sin(ln(1+x))-ln(1+sin(x));
```

$$g := x \to \sin(\ln(1+x)) - \ln(1+\sin(x)) \quad (9)$$

```
> taylor(g(x),x,5);
```

$$\frac{1}{12}x^4 + O(x^5) \quad (10)$$

On a donc $g(x) = \frac{x^4}{12} + o(x^4)$.

```
> taylor(g(x)-x^4/12,x,5);
```

$$O(x^5) \quad (11)$$

Donc si on pose $f(x) = \frac{x^4}{12}$, on a $g(x) - f(x) = O(x^5) = o(x^4) = o\left(\frac{x^4}{12}\right) = o(f(x))$. Donc $g(x)$ est équivalent en 0 à $\frac{x^4}{12}$.

Exercice 3

```
> restart;
```

A l'aide de *seq*, on crée la séquence des parties principales des développements de Taylor d'ordre *2k* pour *k* variant de 1 à 6.

```
> S:=seq(convert(taylor(sin(x),x,2*k),polynom),k=1..6);
```

$$S := x,\ x - \frac{1}{6}x^3,\ x - \frac{1}{6}x^3 + \frac{1}{120}x^5,\ x - \frac{1}{6}x^3 + \frac{1}{120}x^5 - \frac{1}{5040}x^7,\ x - \frac{1}{6}x^3 + \frac{1}{120}x^5 - \frac{1}{5040}x^7 + \frac{1}{362880}x^9,\ x - \frac{1}{6}x^3 + \frac{1}{120}x^5 - \frac{1}{5040}x^7 + \frac{1}{362880}x^9 - \frac{1}{39916800}x^{11} \quad (12)$$

On les fait afficher sur un même graphique, avec $\sin(x)$, et on recadre en ordonnées pour mieux y voir :

```
> plot([sin(x),S],x,y=-2..2,numpoints=1000);
```



Remarquez que les parties principales approchent de mieux en mieux la fonction autour de 0, au fur et à mesure que l'ordre augmente (elles "collent" à la fonction sur un intervalle de plus en plus grand autour de 0). Remarquez aussi que la partie principale du DL à l'ordre mathématique 1 fournit la tangente à la courbe en $x = 0$ (son équation est $y = x$).

<u>Bonus :</u> une version animée !

```
> with(plots):
Warning, the name changecoords has been redefined
```

Avec *seq*, on construit la séquence des graphiques représentant le sinus et la partie principale à l'ordre $2k$. Les *plot* sont des objets comme les autres et on peut les stocker dans une séquence ! Attention cependant à bien mettre deux-points (:) à la fin de la commande (essayez avec point-virgule...)

```
> S:=seq( plot([sin(x),convert(taylor(sin(x),x,2*k),polynom)],x,y=-5..5) ,k=1..10 ):
```

Ensuite, on utilise *display* (librairie *plots*), qui permet d'afficher plusieurs graphiques en même temps. L'option *insequence* = *true* permet de les afficher au fur et à mesure, et donc de faire une animation.

```
> ?display
> display(S,insequence=true);
```

Essayez la commande ci-dessus. Pour démarrer l'animation, cliquez sur le graphique puis sur l'icône de lecture (flèche) qui apparaît en haut. La vitesse peut être réglée par le menu spécial qui apparaît en haut de l'écran.
Pour cos :

```
> T:=seq(convert(taylor(cos(x),x,2*k-1),polynom),k=1..6);
```

$$S := 1,\ 1 - \frac{1}{2}x^2,\ 1 - \frac{1}{2}x^2 + \frac{1}{24}x^4,\ 1 - \frac{1}{2}x^2 + \frac{1}{24}x^4 - \frac{1}{720}x^6,\ 1 - \frac{1}{2}x^2 + \frac{1}{24}x^4 - \frac{1}{720}x^6 + \frac{1}{40320}x^8,\ 1 - \frac{1}{2}x^2 + \frac{1}{24}x^4 - \frac{1}{720}x^6 + \frac{1}{40320}x^8 - \frac{1}{3628800}x^{10} \tag{13}$$

```
> plot([cos(x),T],x,y=-2..2,numpoints=1000);
```




```
> T:=seq( plot([cos(x),convert(taylor(cos(x),x,2*k-1),polynom)],x,y=-5..5) ,k=1..10 ):display(T,insequence=true); # pour la version animée
```

<u>**Exercice 4**</u>

1)

```
> restart;
  f:=x->exp(1/x)*sqrt(x^2+2*x);
```

$$f := x \to e^{\left(\frac{1}{x}\right)} \sqrt{x^2 + 2\,x} \quad (14)$$

```
> taylor(f(x),x=infinity,3);
  taylor(f(x),x=-infinity,3);
```

$$x + 2 + \frac{1}{x} + O\left(\frac{1}{x^2}\right)$$

$$-x - 2 - \frac{1}{x} + O\left(\frac{1}{x^2}\right) \quad (15)$$

Donc la droite d'équation $y = x + 2$ est asymptote à la courbe C en $+\infty$. Au voisinage de $+\infty$, la courbe est située au-dessus de la droite. La droite d'équation $y = -x - 2$ est asymptote à la courbe C en $-\infty$. Au voisinage de $-\infty$, la courbe est située au-dessus de la droite.

```
> plot([f(x),x+2,-x-2],x,y=-10..10,color=[red,blue,green]);
```



2)

```
> g:=x->((x^2-x-6)^2-(x+3))/(x^2-4*x+3);
```

$$g := x \to \frac{\left(x^2 - x - 6\right)^2 - x - 3}{x^2 - 4\,x + 3} \quad (16)$$

```
> taylor(g(x),x=infinity,2);
```

$$x^2 + 2\,x - 6 - \frac{19}{x} + O\left(\frac{1}{x^2}\right) \quad (17)$$

```
> taylor(g(x),x=-infinity,2);
```

$$x^2 + 2\,x - 6 - \frac{19}{x} + O\left(\frac{1}{x^2}\right) \quad (18)$$

Donc la parabole d'équation $y = x^2 + 2\,x - 6$ est asymptote à C en $+\infty$ et $-\infty$. De plus, au voisinage de $+\infty$, la courbe est au-dessous de la parabole ; au voisinage de $-\infty$, la courbe est au-dessus de la parabole.

```
> plot([g(x),x^2+2*x-6],x,y=-20..50,discont=true);
```



3)

```
> h:=x->sqrt(1+x+x^2+x^4);
```

$$h := x \to \sqrt{1 + x + x^2 + x^4} \quad (19)$$

```
> taylor(h(x),x=infinity,2);
  taylor(h(x),x=-infinity,2);
```

$$x^2 + \frac{1}{2} + \frac{1}{2}\,\frac{1}{x} + O\left(\frac{1}{x^2}\right)$$

$$x^2 + \frac{1}{2} + \frac{1}{2}\,\frac{1}{x} + O\left(\frac{1}{x^2}\right) \quad (20)$$

Donc la parabole d'équation $y = x^2 + \frac{1}{2}$ est asymptote à la courbe C en $+\infty$ et $-\infty$. Au voisinage de $+\infty$, la courbe est située au-dessus de la parabole. Au voisinage de $-\infty$, la courbe est située en-dessous de la parabole.

```
> plot([h(x),x^2+1/2],x=-3..3,y=0..10,color=[red,blue]);
```


10.0
7.5
y
5.0
2.5
0.0
−2
0
2
x